ÜBUNG [00]: Wiederholung: Analysis II
Wiederhole die Themen der Analysis aus diesem Semester, wir stellen Fragen im Testat!

Differentation im $\mathbb{R}^n$

```
> restart;
  with(plots):
```

Richtungsableitungen & partielle Differenzierbarkeit

Will man Funktionen $f: \mathrm{D} \rightarrow \mathbb{R}^m$ für Teilmengen $\mathrm{D} \subseteq \mathbb{R}^n$ untersuchen, so hat man das Problem, dass sich die Begriffe des eindimensionalen Falles nicht ohne weiteres übertragen lassen. Will man z.B. Differenzierbarkeit betrachten, so hat man das Problem, dass die Funktion auf einmal von mehr als einer Variable abhängt. Dieses kann man durch die Verkettung mit einer Funktion von $\mathbb{R}$ nach D soweit in den Griff kriegen, dass wir in jeder Komponente eine Funktion des bekannten eindimensionalen Falls haben.

BEISPIEL:

```
> f:=(x,y,z)->x^3*y^2-(x+y)*cos(z);
```

$$f := (x, y, z) \rightarrow x^3 y^2 - (x+y)\cos(z) \quad \textbf{(1.1.1)}$$

sei eine Funktion von $\mathbb{R}^3 \rightarrow \mathbb{R}$.

```
> c:=[t,t^2,t^3];
```

$$c := [t, t^2, t^3] \quad \textbf{(1.1.2)}$$

ist eine Funktion von $\mathbb{R} \rightarrow \mathbb{R}^3$ auch Kurve genannt. Nun bilden wir die Verkettung $f \circ c$:

```
> fc:=f(op(c));
```

$$fc := t^7 - (t^2 + t)\cos(t^3) \quad \textbf{(1.1.3)}$$

Dies ist eine Funktion $\mathbb{R} \rightarrow \mathbb{R}$ und wir können z.b. die Ableitung im Punkte $t = 2$ bestimmen:

```
> diff(fc,t);
  subs(t=2,%);
```

$$7\,t^6 - (2\,t+1)\cos(t^3) + 3\,(t^2+t)\sin(t^3)\,t^2$$

$$448 - 5\cos(8) + 72\sin(8) \quad \textbf{(1.1.4)}$$

So können wir z.B. Wachstumsaussagen über unsere Funktion f entlang der Kurve c treffen. Beachte hierbei, dass der Punkt $t = 2$ dem Vektor

$\begin{bmatrix} 2 \\ 4 \\ 8 \end{bmatrix}$ entspricht.

```
> spacecurve(c,t=-1..1,thickness=3,axes=boxed,color=red);
```

```
animate(spacecurve,[c,t=-1..A,thickness=3,axes=boxed,color=
red],A=-1..1,frames=50);
```





Wie man sieht liegt unsere Kurve "relativ wild" im Raum, man beschränkt sich hier auf die einfachsten Kurven: Geraden.

MATH: Es sei $f\colon \mathrm{D} \to \mathbb{R}^m$ eine Funktion von $\mathrm{D} \subseteq \mathbb{R}^n$ nach $\mathbb{R}^m$. Für einen Vektor $v \in \mathbb{R}^n \setminus \{0\}$ und einen Punkt $a \in \mathrm{D}$ mit der Eigenschaft, dass ein $\rho \in \mathbb{R}_{>0}$ derart existiert, dass $\{a + t \cdot v \mid |t| < \rho\} \subseteq \mathrm{D}$ gilt und

$$\mathrm{D}_v f(a) := \lim_{t \to 0} \frac{1}{t} \left(f(a + t \cdot v) - f(a) \right)$$

existiert, definieren wir diesen Grenzwert als die **Richtungsableitung** von f im Punkt a mit Richtung v. Dies ist äquivalent dazu, dass $f \circ c$ als eindimensionale Funktion im Punkte $t = 0$ differenzierbar ist, wobei c als $c := (-\rho, \rho) \to \mathrm{D} : t \mapsto a + t \cdot v$ gegeben ist.

BEISPIEL: Wir betrachten erneut die Funktion f von oben:

```
> f(x,y,z);
```

$$x^3 y^2 - (x + y) \cos(z) \qquad \textbf{(1.1.5)}$$

Wir können als Richtungen die 3 Standardbasisvektoren nehmen, den Punkt a lassen wir ganz allgemein:

```
> e1,e2,e3:=Vector([1,0,0]),Vector([0,1,0]),Vector([0,0,1]);
  a:=Vector(3,symbol=A);
```

$$e1, e2, e3 := \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}$$

$$a := \begin{bmatrix} A_1 \\ A_2 \\ A_3 \end{bmatrix} \qquad \textbf{(1.1.6)}$$

```
> limit(1/t*(f(op(convert(a+t*e1,list)))-f(op(convert(a,list))
  )),t=0);
  limit(1/t*(f(op(convert(a+t*e2,list)))-f(op(convert(a,list))
  )),t=0);
  limit(1/t*(f(op(convert(a+t*e3,list)))-f(op(convert(a,list))
  )),t=0);
```

$$3\,A_1^2\,A_2^2 - \cos(A_3)$$

$$2\,A_1^3\,A_2 - \cos(A_3)$$

$$\sin(A_3)\,A_1 + \sin(A_3)\,A_2 \qquad \textbf{(1.1.7)}$$

MAPLE: Leider können wir keine Vektoren in die Funktion f direkt einsetzen, daher müssen wir diese immer zuerst in Listen umwandeln und mittels **op** die Klammern entfernen. Alternativ können wir die Funktion anders in Maple definieren, um Vektoren einsetzen zu können:

```
> ff:= v -> v[1]^3*v[2]^2-(v[1]+v[2])*cos(v[3]);
```

$$ff := v \to v_1^3\,v_2^2 - (v_1 + v_2)\cos(v_3) \qquad \textbf{(1.1.8)}$$

```
> limit(1/t*(ff(a+t*e1)-ff(a)),t=0);
```

$$3\,A_1^2\,A_2^2 - \cos(A_3) \qquad \textbf{(1.1.9)}$$

Leider ist diese Methode auch nicht optimal:

```
> ff(Vector([1,2]));
Error, (in ff) Vector index out of range
```

ÜBUNG [01]:

Bestimme Richtungsableitungen $\mathrm{D}_v f(a)$ für einen allgemeinen Punkt a und die Richtungen $v \in \left\{ \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}, \begin{bmatrix} 2 \\ 3 \\ 4 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix} \right\}$:

```
f:=(x,y,z)->exp(x^2+y)*z^2-sin(x)*cos(y);
```

$$f := (x, y, z) \to \mathrm{e}^{x^2 + y}\,z^2 - \sin(x)\cos(y) \qquad \textbf{(1.1.10)}$$

Welcher Zusammenhang fällt dir auf?

Wir stellen fest, dass der Begriff Richtungsableitung ein wenig irreführend ist, da es eine Abhängigkeit von der Geschwindigkeit, mit der die Kurve durchlaufen wird, gibt:

```
> v,w:=Vector([1,2,3]),Vector([3,6,9]);
```

$$v, w := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}, \begin{bmatrix} 3 \\ 6 \\ 9 \end{bmatrix} \tag{1.1.11}$$

```
> D_v_f:=limit(1/t*(ff(a+t*v)-ff(a)),t=0);
```

$$D\_v\_f := 2\sin(A_1)\sin(A_2) - \cos(A_1)\cos(A_2) + 6\,e^{A_1^2}\,e^{A_2}\,A_3 + 2\,e^{A_1^2}\,e^{A_2}\,A_3^2 + 2\,e^{A_2}\,A_3^2\,A_1\,e^{A_1^2} \tag{1.1.12}$$

```
> D_w_f:=limit(1/t*(ff(a+t*w)-ff(a)),t=0);
```

$$D\_w\_f := 6\sin(A_1)\sin(A_2) - 3\cos(A_1)\cos(A_2) + 18\,e^{A_1^2}\,e^{A_2}\,A_3 + 6\,e^{A_1^2}\,e^{A_2}\,A_3^2 + 6\,e^{A_2}\,A_3^2\,A_1\,e^{A_1^2} \tag{1.1.13}$$

```
> v*3=w;
  is(D_v_f*3=D_w_f);
```

$$\begin{bmatrix} 3 \\ 6 \\ 9 \end{bmatrix} = \begin{bmatrix} 3 \\ 6 \\ 9 \end{bmatrix}$$

$$\mathit{true} \tag{1.1.14}$$

ÜBUNG [02]:
Beweise die obige Beobachtung: Für $\lambda \neq 0$ gilt: $\mathrm{D}_{\lambda v} f(a) = \lambda \cdot \mathrm{D}_v f(a)$.

Unter den Richtungsableitungen sind die der Standardbasisvektoren ausgezeichnet:

MATH: Sei $\mathrm{D} \subseteq \mathbb{R}^n$ offen, $a = \begin{bmatrix} a_1 \\ \ldots \\ a_n \end{bmatrix} \in \mathrm{D}$ und $f: \mathrm{D} \to \mathbb{R}^m$ eine Funktion. Wir nennen f in a **partiell differenzierbar bezüglich der *k*-ten Koordinate** $(1 \leq k \leq n)$, wenn $\mathrm{D}_k f(a) := \frac{\partial f}{\partial x_k}(a) := \mathrm{D}_{e_k} f(a) \in \mathbb{R}^m$ existiert. Die Funktion f heißt **partiell**

differenzierbar, wennn $\mathrm{D}_k f(a)$ für alle $a \in \mathrm{D}$ und alle $k \in \{1,\dots, n\}$ existiert. Die Abbildungen $\mathrm{D}_k f : \mathrm{D} \to \mathbb{R}^m$ heißen **partielle Ableitungen** von f. Falls alle partiellen Ableitungen auf D existieren, so definieren wir $Df(x) := (\mathrm{D}_1 f(x),\dots \mathrm{D}_n f(x)) \in \mathbb{R}^{m \times n}$ für $x \in \mathrm{D}$. Diese Matrix $Df(x)$ heißt **Jacabi-Matrix** oder auch **Differential** von f an der Stelle x. Ferner heißt f **stetig partiell differenzierbar**, falls f partiell differenzierbar ist und die partiellen Ableitungen $\mathrm{D}_k f$ alle stetig sind.

MAPLE: kennt natürlich die Jacobimatrix und kann diese ausrechnen:

```
> Df:=VectorCalculus[Jacobian]([f(x,y,z)],[x,y,z]);
  LinearAlgebra[Dimensions](Df);
  Df[1,2];
```

$$Df := \begin{bmatrix} 2\,x\,\mathrm{e}^{x^2+y}\,z^2 - \cos(x)\cos(y) & \mathrm{e}^{x^2+y}\,z^2 + \sin(x)\sin(y) & 2\,\mathrm{e}^{x^2+y}\,z \end{bmatrix}$$

$$1,\ 3$$

$$\mathrm{e}^{x^2+y}\,z^2 + \sin(x)\sin(y) \tag{1.1.15}$$

Wie erwartet erhalten wir eine 1×3 Matrix. Alternativ können wir auch den Befehl aus dem Paket laden, um die Kurzform zu verwenden:

```
> with(VectorCalculus,Jacobian):
> Jacobian([f(x,y,z)],[x,y,z]);
```

$$\begin{bmatrix} 2\,x\,\mathrm{e}^{x^2+y}\,z^2 - \cos(x)\cos(y) & \mathrm{e}^{x^2+y}\,z^2 + \sin(x)\sin(y) & 2\,\mathrm{e}^{x^2+y}\,z \end{bmatrix} \tag{1.1.16}$$

MATH: Sei $f : \mathrm{D} \to \mathbb{R}^m$ mit $\mathrm{D} \subseteq \mathbb{R}^n$ eine Funktion, $a \in \mathrm{D}$ ein Punkt und $\delta > 0$ so, dass $\{a + t \cdot e_k \mid |t| < \delta\} \subseteq \mathrm{D}$ gilt. Dann ist f in a genau dann bezüglich der k-ten Koordinate partiell differenzierbar, falls die Funktion $\varphi_k : (a_k - \delta, a_k + \delta) \to \mathbb{R} : t \mapsto f((a_1,\dots, a_{k-1}, t, a_{k+1},\dots, a_n))$ in a_k differenzierbar ist. Dann gilt: $\mathrm{D}_k f(a) = \varphi_k'(a_k)$.

ÜBUNG [03]:

Untersuche die 2-Norm $\|\ \|_2 : \mathbb{R}^n \to \mathbb{R} : v \mapsto \sqrt{\sum_{i=1}^{n} v_i^2}$ als Abbildung des $\mathbb{R}^n$ auf partielle Differenzierbarkeit.
Hinweis: Nicht alle Punkte sind gleich, manche sind gleicher.

Analog zum bekannten aus Analysis I definiert man mehrfache partielle Differenzierbarkeit:

MATH: Sei $f : \mathrm{D} \to \mathbb{R}^m$ mit $\mathrm{D} \subseteq \mathbb{R}^n$ offen. Wir nennen f **zweimal partiell differenzierbar**, wenn f partiell differenzierbar ist und alle partiellen Ableitungsfunktionen $\mathrm{D}_i \mathrm{D}_j f := \mathrm{D}_i(\mathrm{D}_j f) := \frac{\partial}{\partial x_i}\left(\frac{\partial}{\partial x_j} f\right) =: \frac{\partial^2}{\partial x_i \partial x_j} f$ für $1 \le i, j \le n$

existieren.

DENKANSTOSS: Definiere die Begriffe k-mal partiell differenzierbar sowie k-mal stetig partiell differenzierbar.

BEISPIEL: Wir betrachten erneut die Funktion f:

```
> f(x,y,z);
  Df;
```

$$e^{x^2+y}\,z^2-\sin(x)\cos(y)$$

$$\begin{bmatrix} 2\,x\,e^{x^2+y}\,z^2-\cos(x)\cos(y) & e^{x^2+y}\,z^2+\sin(x)\sin(y) & 2\,e^{x^2+y}\,z \end{bmatrix} \qquad \textbf{(1.1.17)}$$

```
> D1D1f:=diff(Df[1,1],x);
  D2D1f:=diff(Df[1,1],y);
  D3D1f:=diff(Df[1,1],z);
```

$$D1D1f := 2\,e^{x^2+y}\,z^2+4\,x^2\,e^{x^2+y}\,z^2+\sin(x)\cos(y)$$

$$D2D1f := 2\,x\,e^{x^2+y}\,z^2+\cos(x)\sin(y)$$

$$D3D1f := 4\,x\,e^{x^2+y}\,z \qquad \textbf{(1.1.18)}$$

```
> D1D2f:=diff(Df[1,2],x);
  D2D2f:=diff(Df[1,2],y);
  D3D2f:=diff(Df[1,2],z);
```

$$D1D2f := 2\,x\,e^{x^2+y}\,z^2+\cos(x)\sin(y)$$

$$D2D2f := e^{x^2+y}\,z^2+\sin(x)\cos(y)$$

$$D3D2f := 2\,e^{x^2+y}\,z \qquad \textbf{(1.1.19)}$$

```
> D1D3f:=diff(Df[1,3],x);
  D2D3f:=diff(Df[1,3],y);
  D3D3f:=diff(Df[1,3],z);
```

$$D1D3f := 4\,x\,e^{x^2+y}\,z$$

$$D2D3f := 2\,e^{x^2+y}\,z$$

$$D3D3f := 2\,e^{x^2+y} \qquad \textbf{(1.1.20)}$$

```
> is(D1D2f=D2D1f);
  is(D1D3f=D3D1f);
  is(D2D3f=D3D2f);
```

true

true

true **(1.1.21)**

Die partiellen Ableitungen scheinen vertauschbar zu sein. Wir betrachten ein warnendes Beispiel:

ÜBUNG [04]:

Bestimme die zweiten partiellen Ableitungen im Ursprung der folgenden

Funktion:

$$f\colon \mathbb{R}^2 \to \mathbb{R}^2 : \begin{bmatrix} x \\ y \end{bmatrix} \mapsto \begin{cases} xy\dfrac{x^2-y^2}{x^2+y^2} & falls \begin{bmatrix} x \\ y \end{bmatrix} \neq \begin{bmatrix} 0 \\ 0 \end{bmatrix} \\ 0 & sonst \end{cases}$$

Was fällt dir auf?

Wann die zweiten partiellen Ableitungen miteinander vertauschbar sind, sagt uns der Satz von Schwarz:

MATH: Sei $\mathrm{D} \subseteq \mathbb{R}^n$ offen und $f\colon \mathrm{D} \to \mathbb{R}^m$ zweimal stetig partiell differenzierbar. Dann gilt:

$$\mathrm{D}_i \mathrm{D}_j f = \mathrm{D}_j \mathrm{D}_i f \quad \text{für alle } 1 \leq i, j \leq n.$$

DENKANSTOSS: Es sei $f\colon \mathrm{D} \to \mathbb{R}^m$ k-mal partiell differenzierbar, $(i_1, \ldots, i_k) \in \{1, .., n\}^k$ und $\pi \in S_k$ eine Permutation. Welche Voraussetzung garantiert uns die Gleichheit: $\mathrm{D}_{i_k} \mathrm{D}_{i_{k-1}} \ldots \mathrm{D}_{i_1} f = \mathrm{D}_{i_{\pi(k)}} \mathrm{D}_{i_{\pi(k-1)}} \ldots \mathrm{D}_{i_{\pi(1)}} f$?

Totale Differenzierbarkeit

Wir erinnern uns an die Differentation im eindimensionalen Fall: Eine Funktion $f\colon \mathbb{R} \to \mathbb{R}$ ist differenzierbar im Punkt a genau dann, wenn die Funktion $\dfrac{f(x) - f(a)}{x-a} - c$ für $c \in \mathbb{R}$ im Punkte a stetig ergänzbar ist. Den Wert c bei welchem die Ergänzung zu 0 erfolgt, war dann $f'(a)$. Das Setzen von $\varphi(x) := \dfrac{f(x) - f(a)}{x-a} - f'(a)$ liefert folgende äquivalente Formulierung: Die Funktion f ist in a genau dann differenzierbar, wenn es eine in a stetige Funktion $\varphi : \mathbb{R} \to \mathbb{R}$ und ein $c \in \mathbb{R}$ gibt, so dass $f(x) = f(a) + c \cdot (x-a) + (x-a) \cdot \varphi(x)$ für $x \in \mathbb{R}$ und $\varphi(a) = 0$ gilt. So inspiriert formulieren wir nun:

MATH: Sei $f\colon \mathrm{D} \to \mathbb{R}^m$ eine Funktion von $\mathrm{D} \subseteq \mathbb{R}^n$ offen, mit $a \in \mathrm{D}$. Dann heißt f **total differenzierbar in** a, wenn es eine Matrix $T \in \mathbb{R}^{m \times n}$ und eine in a stetige Funktion $\varphi : \mathrm{D} \to \mathbb{R}^m$ gibt, mit der Eigenschaft, dass

$$f(x) = f(a) + T \cdot (x-a) + \|x-a\| \cdot \varphi(x) \text{ für alle } x \in \mathrm{D} \text{ und } \varphi(a) = 0 \text{ gilt.}$$

Wir nennen f **total differenzierbar**, wenn f in allen Punkten $a \in \mathrm{D}$ total differenzierbar ist.

Wir können auch hier nach $\varphi(x)$ auflösen und erhalten die äquivalente Formulierung:

MATH: Sei f wie oben. Dann ist f genau dann total differenzierbar, wenn es ein $T \in \mathbb{R}^{m \times n}$ gibt, dass $\lim_{x \to a} \frac{1}{\|x-a\|} (f(x) - f(a) - T(x-a)) = 0$ gilt.

BEISPIEL: Das einfachste Beispiel ist eine lineare Abbildung gegeben durch die Matrix A.

```
> A:=Matrix([[1,2,3],[2,3,4]]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 2 & 3 & 4 \end{bmatrix} \tag{1.2.1}$$

```
> f:=(x,y,z)->A.Vector([x,y,z]);
```

$$f := (x, y, z) \rightarrow A.Vector([x, y, z]) \tag{1.2.2}$$

```
> f(x,y,z)-A.Vector([x,y,z]);
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} \tag{1.2.3}$$

Wir sehen also sofort, dass $\varphi = 0$ die geforderte Abbildung ist.

Wir wollen uns aber nicht unnötig lange mit Beispielen aufhalten, denn es gibt ein sehr einfaches Kriterium, welches den Zusammenhang zur partiellen Differenzierbarkeit herstellt:

MATH: Sei f wie oben und total differenzierbar in a. Dann ist f in a partiell differenzierbar. Ferner ist die Matrix T eindeutig bestimmt und es gilt $T = (Df)(a)$. Weiter gilt für die Richtungsableitungen mit $v \in \mathbb{R}^n \setminus \{0\}$: $\mathrm{D}_v f(a) = Df(a) \cdot v$.

Totale Differenzierbarkeit ist aber nicht äquivalent zur partiellen, wir müssen etwas mehr fordern:

MATH: Sei f wie oben, partiell differenzierbar und stetig partiell differenzierbar in a. Dann ist f in a total differenzierbar (und damit insbesondere auch stetig).

ÜBUNG [05]:

Zeige, dass die folgende Funktion $\mathbb{R}^2 \rightarrow \mathbb{R}^2$ im Punkt $(1, 1)$ total differenzierbar ist, indem du
1) Mit Maple zeigst, dass sie stetig partiell differenzierbar ist.
2) Per Hand die Definition der totalen Differenzierbarkeit überprüfst.

```
> g:=(x,y)->[x+y^2,x^3+5*y];
```

$$g := (x, y) \rightarrow [x + y^2, x^3 + 5\,y] \tag{1.2.4}$$

Auch im mehrdimensionalen Fall gilt die Kettenregel:

MATH: Seien $U \subseteq \mathbb{R}^n$ offen, $V \subseteq \mathbb{R}^m$ offen, $f\colon U \to \mathbb{R}^m$ total differenzierbar in $a \in U$ mit $f(U) \subseteq V$ sowie $g\colon V \to \mathbb{R}^o$ total differenzierbar in $b = f(a)$. Dann ist $h := g \circ f\colon U \to \mathbb{R}^o$ total differenzierbar in a mit

$$(\mathrm{D}(g \circ f))(a) = (Dg)(f(a)) \cdot (Df)(a).$$

BEISPIEL:

```
> f:=[x*y^2*z^3,(y+x^2)^4];
```

$$f := \left[x\,y^2\,z^3,\ (x^2+y)^4\right] \tag{1.2.5}$$

```
> DF:=Jacobian(f,[x,y,z]);
```

$$DF := \begin{bmatrix} y^2\,z^3 & 2\,x\,y\,z^3 & 3\,x\,y^2\,z^2 \\ 8\,(x^2+y)^3\,x & 4\,(x^2+y)^3 & 0 \end{bmatrix} \tag{1.2.6}$$

```
> g:=[x+y^2,x^3+5*y];
```

$$g := \left[y^2+x,\ x^3+5\,y\right] \tag{1.2.7}$$

```
> DG:=Jacobian(g,[x,y]);
```

$$DG := \begin{bmatrix} 1 & 2\,y \\ 3\,x^2 & 5 \end{bmatrix} \tag{1.2.8}$$

```
> gf:=subs([x=f[1],y=f[2]],g);
```

$$gf := \left[(x^2+y)^8 + x\,y^2\,z^3,\ x^3\,y^6\,z^9 + 5\,(x^2+y)^4\right] \tag{1.2.9}$$

```
> DGF:=Jacobian(gf,[x,y,z]);
```

$$DGF := \begin{bmatrix} 16\,(x^2+y)^7\,x + y^2\,z^3 & 8\,(x^2+y)^7 + 2\,x\,y\,z^3 & 3\,x\,y^2\,z^2 \\ 3\,x^2\,y^6\,z^9 + 40\,(x^2+y)^3\,x & 6\,x^3\,y^5\,z^9 + 20\,(x^2+y)^3 & 9\,x^3\,y^6\,z^8 \end{bmatrix} \tag{1.2.10}$$

```
> a:=[1,2,3];
```

$$a := [1, 2, 3] \tag{1.2.11}$$

```
> DGF_a:=subs([x=1,y=2,z=3],DGF);
```

$$DGF\_a := \begin{bmatrix} 35100 & 17604 & 108 \\ 3780216 & 3779676 & 3779136 \end{bmatrix} \tag{1.2.12}$$

```
> subs([x=1,y=2,z=3],f);#f(1,2,3)
  DG_fa_DF_a:=subs([x=108,y=81],DG).subs([x=1,y=2,z=3],DF);
```

$$[[108, 81]]$$

$$DG\_fa\_DF\_a := \begin{bmatrix} 35100 & 17604 & 108 \\ 3780216 & 3779676 & 3779136 \end{bmatrix} \tag{1.2.13}$$

```
> LinearAlgebra[Equal](DG_fa_DF_a,DGF_a);
```

$$\mathit{true} \tag{1.2.14}$$

ÜBUNG [06]:

1) Differenziere $x \mapsto x^x$ mit Hilfe der Kettenregel, indem diese Funktion als

Komposition von

$$x \mapsto (x, x)$$

und

$$(x, y) \mapsto ?$$

geschrieben wird.

2) Leite (mit derselben Idee) die eindimensionale Produktregel aus der Kettenregel her.

Motivation: Implizite Funktionen

MATH: Abbildungen

$$g: \mathbb{R}^n \to \mathbb{R}$$

visualisiert man im Falle $n = 2$ durch ihre Graphen.

```
> plot3d(x^2-y^3,x=-1..1,y=-1..1, axes=FRAME);
```



MATH: Für Abbildungen

$g : \mathbb{R}^n \to \mathbb{R}$

kann man auf vielerlei Weisen von Wachstum sprechen: Jede Kurve

$c : \mathbb{R} \to \mathbb{R}^n$

liefert uns durch Einsetzen eine Funktion

$g \circ c : \mathbb{R} \to \mathbb{R} : t \mapsto g(c(t))$

wie wir sie schon immer studiert haben.

Hier sind zwei Möglichkeiten, dies in MAPLE zu realisieren:

```
> g:=x[1]^2+x[2]-x[3]^3;
```

$$g := -x_3^3 + x_1^2 + x_2 \qquad (1.3.1)$$

```
> c;
```

$$[t, t^2, t^3] \qquad (1.3.2)$$

```
> subs(zip((i,j)->x[i]=j,[$1..3],c),g);
```

$$-t^9 + 2\,t^2 \qquad (1.3.3)$$

```
> h:=(x,y,z)->x^2+y-z^3;
```

$$h := (x, y, z) \to x^2 + y - z^3 \qquad (1.3.4)$$

```
> h(op(c));
```

$$-t^9 + 2\,t^2 \qquad (1.3.5)$$

Nun fragt man sich: Für welche Kurven c ist die Komposition konstant, also ohne Wachstum? ("Höhenlinien" in Graphen)

MATH: Diese Frage läuft darauf hinaus, die Fasern der Abbildung g zu bestimmen, also die Gleichungssysteme

$g(x_1, \ldots, x_n) = k$

für jedes feste $k \in \mathbb{R}$ zu lösen. Die Bedingung an die Kurve c ist dann einfach nur, dass sie ihre Werte in einer festen Faser

$g^{-1}(\{k\})$

annimmt oder, geometrisch gesprochen, in einer festen Faser verläuft.

Wir betrachten die Faser $g^{-1}(\{0\})$ als Teilmenge des $\mathbb{R}^3$:

```
> implicitplot3d(x^2+y-z^3=0,x=-1..1,y=-1..1,z=-1..1,axes=
  boxed);
```



Ein besseres Bild bekommt man, wenn man die Gleichung nach *y* auflöst, weil man dann die Faser als Graph einer Funktion darstellen kann. (Diese Auflösung ist im vorliegenden Fall sehr einfach, im Allgemeinen aber nicht immer möglich.) Wir stellen nicht nur die Faser über 0, sondern auch die über $\frac{1}{2}$ und $-\frac{1}{2}$ dar, damit die Faserung des $\mathbb{R}^3$ durch f augenfälliger wird:

```
> plot3d(map(a->a+z^3-x^2,{-1/2,0,1/2}),x=-1..1,z=-1..1,axes=
  boxed);
```



Diese beiden Bilder sind nicht direkt vergleichbar. Das wollen wir beheben.

ÜBUNG [07]:
1) Modifiziere die Reihenfolge der Achsen bei den beiden obigen Plot-Befehlen so, dass die Bilder direkt vergleichbar werden. (Hinweis: Schreibe zur Kontrolle **B1:= implicitplot3d(....): B2:= plot3d(....): display({B1,B2});**) 2) Gib eine Kurve im $\mathbb{R}^3$ an, die ganz in der Faser von g über 0 verläuft.

Man kann zeigen, dass man (unter milden Voraussetzungen) immer (zumindest lokal) die Faser als Funktion auffassen kann.